// File: Contracts/Rewarder.sol



// SPDX-License-Identifier: MIT

pragma solidity 0.8.16;

import { IRewarder } from "./interfaces/IRewarder.sol";
import { IRewarderVault } from "./interfaces/IRewarderVault.sol";
import { IMasterMind } from "./interfaces/IMasterMind.sol";

import { Ownable } from "@openzeppelin/contracts/access/Ownable.sol";
import { IERC20, SafeERC20 } from "@openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol";

contract Rewarder is IRewarder, Ownable {
    using SafeERC20 for IERC20;

    bool private _migrated;
    IERC20 public immutable rewardToken;
    IMasterMind public immutable masterMind;
    IRewarderVault public immutable rewarderVault;

    mapping (uint256 => PoolInfo) public poolInfo;
    mapping (uint256 => mapping (address => UserInfo)) public userInfo;
    uint256 private MAX_REWARD_PER_BLOCK_PER_POOL = 100e18;

    address public dev;
    address public dao;
    uint256 public constant FEE_BASE = 1000;
    uint256 public devFee = 100;
    uint256 public daoFee = 100;
    uint256 public devDebt;
    uint256 public daoDebt;

    constructor (IERC20 _rewardToken, IMasterMind _masterMind, IRewarderVault _rewarderVault, address _dao) {
        require(address(_rewardToken) != address(0) && address(_masterMind) != address(0) && address(_rewarderVault) != address(0), "Wrong Init");
        rewardToken = _rewardToken;
        masterMind = _masterMind;
        rewarderVault = _rewarderVault;
        dev = msg.sender;
        dao = _dao;
    }

    modifier onlyMM {
        require(
            msg.sender == address(masterMind),
            "Only MM can call this function."
        );
        _;
    }

    function setDevFee(uint256 value_) external onlyOwner {
        require(value_ <= 350, 'too much');
        devFee = value_;
    }

    function setDaoFee(uint256 value_) external onlyOwner {
        require(value_ <= 350, 'too much');
        daoFee = value_;
    }

    function updateMaxRewardPerBlockPerPool(uint256 amount_) external onlyOwner {
        MAX_REWARD_PER_BLOCK_PER_POOL = amount_;
    }

    struct UserMigration {
        address address_;
        UserInfo info;
    }

    struct MigrationPackage {
        uint256 poolId;
        PoolInfo poolInfo;
        UserMigration[] users;
    }

    function migrate(MigrationPackage[] calldata packs) external onlyOwner {
        require(!_migrated);
        uint256 l = packs.length;
        for (uint256 i; i < l;){
            MigrationPackage memory pack = packs[i];
            uint256 poolId = pack.poolId;
            poolInfo[poolId] = pack.poolInfo;
            uint256 ul = pack.users.length;
            for (uint256 j; j < ul;) {
                UserMigration memory user = pack.users[j];
                userInfo[poolId][user.address_] = user.info;
            unchecked{ ++j; }
            }
        unchecked{ ++i; }
        }
        _migrated = true;
    }

    function updateUser(uint256 poolId, address userAddress, uint256 rewardableDeposit) external override onlyMM {
        (PoolInfo memory pool,) = updatePool(poolId);
        UserInfo storage user = userInfo[poolId][userAddress];
        assert(pool.lifetimeRewardPerOneEtherOfDeposit >= user.lifetimeRewardPerOneEtherOfDeposit);
        uint256 oneEtherOfDepositRewardAppreciation;
    unchecked {
        oneEtherOfDepositRewardAppreciation = pool.lifetimeRewardPerOneEtherOfDeposit - user.lifetimeRewardPerOneEtherOfDeposit;
    }
        if(oneEtherOfDepositRewardAppreciation > 0) {
            uint256 rewardAmount = oneEtherOfDepositRewardAppreciation * user.rewardableDeposit;
            user.pendingReward = user.pendingReward + (rewardAmount / 1 ether);
        }

        user.rewardableDeposit = rewardableDeposit;
        user.lifetimeRewardPerOneEtherOfDeposit = pool.lifetimeRewardPerOneEtherOfDeposit;
        emit UpdateUser(userAddress, poolId, rewardableDeposit);
    }

    function claim(uint256 poolId, address userAddress, address to) onlyMM override external {
        UserInfo storage user = userInfo[poolId][userAddress];
        uint256 rewardAmount = user.pendingReward;
        if(rewardAmount > 0) {
            user.pendingReward = 0;
            rewardToken.safeTransferFrom(address(rewarderVault), to, rewardAmount);
            devDebt += rewardAmount*devFee / FEE_BASE;
            daoDebt += rewardAmount*daoFee / FEE_BASE;
            emit Claim(userAddress, poolId, rewardAmount, to);
        }
    }

    /// @notice Returns the number of MM pools.
    function poolCount() public view override returns (uint256 pools) {
        pools = masterMind.poolCount();
    }

    function addBulk(uint256[] memory rewardsPerBlockPerOneEther, uint256[] memory poolIdList, uint256[] memory poolLimits) external override onlyOwner{
        for (uint i = 0; i < rewardsPerBlockPerOneEther.length; i++) {
            uint256 rewardPerBlockPerOneEther = rewardsPerBlockPerOneEther[i];
            uint256 poolId = poolIdList[i];
            require(poolInfo[poolId].lastRewardBlock == 0, "Pool already exists");
            poolInfo[poolId] = PoolInfo({
            rewardPerBlockPerOneEtherOfDeposit: rewardPerBlockPerOneEther,
            lastRewardBlock: block.number,
            lifetimeRewardPerOneEtherOfDeposit: 0,
            limitPerBlockPerOneEtherOfDeposit: poolLimits[i]
            });

            emit AddPool(poolId, rewardsPerBlockPerOneEther[i]);
        }
    }

    function updatePoolRewards(uint256 poolId, uint256 newRewardPerBlockPerEther) external override onlyOwner {
        _updatePoolRewards(poolId, newRewardPerBlockPerEther);
        emit UpdatePoolRewards(poolId, newRewardPerBlockPerEther);
    }

    function massUpdatePoolRewards(uint256[] memory poolIds, uint256[] memory newRewardsPerBlockPerEther) external override onlyOwner {
        require(poolIds.length == newRewardsPerBlockPerEther.length, "non-equal arr length");
        for (uint i = 0; i < poolIds.length; i++) {
            _updatePoolRewards(poolIds[i], newRewardsPerBlockPerEther[i]);
        }
        emit UpdatePoolsRewards(poolIds, newRewardsPerBlockPerEther);
    }

    function _updatePoolRewards(uint256 poolId, uint256 newRewardPerBlockPerEther) internal {
        (PoolInfo memory pool, uint256 rewardableDeposits) = updatePool(poolId);
        require(newRewardPerBlockPerEther < pool.limitPerBlockPerOneEtherOfDeposit &&
            rewardableDeposits * pool.limitPerBlockPerOneEtherOfDeposit < MAX_REWARD_PER_BLOCK_PER_POOL, "Rewards are too big.");
        pool.rewardPerBlockPerOneEtherOfDeposit = newRewardPerBlockPerEther;
        poolInfo[poolId] = pool;
    }

    function updatePoolLimit(uint256 poolId, uint256 newLimitPerBlockPerEther) external override onlyOwner {
        _updatePoolLimit(poolId, newLimitPerBlockPerEther);
    }

    function massUpdatePoolLimits(uint256[] memory poolIds, uint256[] memory newLimitsPerBlockPerEther) external override onlyOwner {
        require(poolIds.length == newLimitsPerBlockPerEther.length, "non-equal arr length");
        for (uint i = 0; i < poolIds.length; i++) {
            _updatePoolLimit(poolIds[i], newLimitsPerBlockPerEther[i]);
        }
    }

    function _updatePoolLimit(uint256 poolId, uint256 newLimitPerBlockPerEther) internal {
        poolInfo[poolId].limitPerBlockPerOneEtherOfDeposit = newLimitPerBlockPerEther;
        emit UpdatePoolLimit(poolId, newLimitPerBlockPerEther);
    }

    function pendingReward(uint256 poolId, address userAddress) external view override returns (uint256 _pendingReward) {
        PoolInfo memory pool = poolInfo[poolId];
        UserInfo storage user = userInfo[poolId][userAddress];
        _pendingReward = user.pendingReward;
        uint256 lifetimeRewardPerOneEtherOfDeposit = pool.lifetimeRewardPerOneEtherOfDeposit;
        if (block.number > pool.lastRewardBlock) {
            uint256 rewardableDeposits = masterMind.poolInfo(poolId).rewardableDeposits;
            if (rewardableDeposits > 0) {
                uint256 blocks = uint256(block.number - pool.lastRewardBlock);
                uint256 rewardPerOneEtherOfDeposit = blocks * poolInfo[poolId].rewardPerBlockPerOneEtherOfDeposit;
                lifetimeRewardPerOneEtherOfDeposit = lifetimeRewardPerOneEtherOfDeposit + rewardPerOneEtherOfDeposit;
            }
        }

        uint256 oneEtherOfDepositRewardAppreciation = lifetimeRewardPerOneEtherOfDeposit - user.lifetimeRewardPerOneEtherOfDeposit;
        if(oneEtherOfDepositRewardAppreciation > 0) {
            uint256 rewardAmount = oneEtherOfDepositRewardAppreciation * user.rewardableDeposit;
            _pendingReward = _pendingReward + (rewardAmount / 1 ether);
        }
    }

    function massUpdatePools(uint256[] calldata poolIdList) external override {
        for (uint256 i = 0; i < poolIdList.length; ++i) {
            updatePool(poolIdList[i]);
        }
    }

    function updatePool(uint256 poolId) public override returns (PoolInfo memory pool, uint256 rewardableDeposits) {
        pool = poolInfo[poolId];
        require(pool.lastRewardBlock != 0, "Pool does not exist");
        rewardableDeposits = masterMind.poolInfo(poolId).rewardableDeposits;
        if (block.number > pool.lastRewardBlock) {
            if (rewardableDeposits > 0) {
                uint256 rewardPerOneEtherOfDeposit = uint256(block.number - pool.lastRewardBlock) * pool.rewardPerBlockPerOneEtherOfDeposit;
                pool.lifetimeRewardPerOneEtherOfDeposit = pool.lifetimeRewardPerOneEtherOfDeposit + rewardPerOneEtherOfDeposit;
            }
            pool.lastRewardBlock = block.number;
            poolInfo[poolId] = pool;
            emit UpdatePool(poolId, pool.lastRewardBlock, rewardableDeposits, pool.lifetimeRewardPerOneEtherOfDeposit);
        }
    }

    function distributeFee() external override {
        require(msg.sender == dev || msg.sender == dao || msg.sender == owner(), "You can't do it");
        uint256 _devDebt = devDebt;
        uint256 _daoDebt = daoDebt;
        daoDebt = 0;
        devDebt = 0;
        rewardToken.safeTransferFrom(address(rewarderVault), dev, _devDebt);
        rewardToken.safeTransferFrom(address(rewarderVault), dao, _daoDebt);
    }

    function updateDev(address newDev) external override onlyOwner {
        dev = newDev;
        emit UpdateDev(newDev);
    }

    function updateDao(address newDao) external override onlyOwner {
        dao = newDao;
        emit UpdateDao(newDao);
    }
}

// File: Contracts/interfaces/IRewarderVault.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

interface IRewarderVault {
    event UpdateGuardian(address newGuard);
    event UpdateRewarder(address newRewarder);
    function rewarder() external returns(address);
    function setGuardian(address newGuardian) external;
    function lock(bool flag) external;
    function updateRewarder(address newRewarder) external;
    function fillVault(uint256 amount) external;
}

// File: Contracts/interfaces/IMasterMind.sol



//SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import { ITarget } from "../libraries/Adapter.sol";
import { Rewarder } from "../libraries/SafeRewarder.sol";

interface IMasterMind {
    struct UserInfo {
        uint256 rewardableDeposit;
        uint256 lastDepositBlock;
    }

    struct PoolInfo {
        ITarget target;
        address adapter;
        uint256 targetPoolId; // recompoundingDeposits should be calculated as lockedAmount(pool) - pool.rewardableDeposits
        uint256 rewardableDeposits;
    }

    struct RolesSlot {
        address owner;
        address service;
    }
    
    struct DrainProtectionSlot {
        mapping (address => bool) allowedDrainers;
        bool enabled;
    }

    struct DrainAddressSlot {
        address value;
    }

    struct RewarderSlot {
        Rewarder value;
    }

    struct UserInfoSlot {
        mapping (uint256 => mapping (address => UserInfo)) users;
    }

    struct PoolInfoSlot {
        mapping(uint256 => PoolInfo) pools;
        uint256 count;
    }

    event  Deposit(address indexed user, uint256 indexed poolId, uint256 rewardableAmount);
    event Withdraw(address indexed user, uint256 indexed poolId,  uint256 rewardableAmount);
    event Drain(uint256 indexed poolId, address indexed token, uint256 claimedAmount);
    event Add(uint256 indexed poolId);
    event AddBulk(uint256 indexed start, uint256 indexed finish);
    event UpdateTargetInfo(uint256 indexed poolId);
    event UpdateAdapter(uint256 indexed poolId);
    
    /* Views */
    function owner() external view returns (address);
    function service() external view returns (address);
    function drainProtectionEnabled() external view returns (bool);
    function allowedDrainer(address drainer) external view returns (bool);
    function drainAddress() external view returns (address);
    function rewarder() external view returns (address);
    function userInfo(uint256 poolId, address user) external view returns (UserInfo memory);
    function poolInfo(uint256 poolId) external view returns (PoolInfo memory);
    function poolCount() external view returns (uint256);
    function lockableToken(uint256 poolId) external view returns(address);
    function userDeposits(uint256 poolId, address userAddress) external view returns (uint256 rewardableDeposit);
    function earnedRewards(uint256 poolId, address userOfTarget) external view returns (address[] memory, uint256[] memory);
    /* Mutators */
    function init(address newOwner) external;
    function updateService(address newService) external;
    function setDrainProtection(bool enabled) external;
    function setDrainerAllowance(address drainer, bool allow) external;
    function updateRewarder(address newRewarder) external;
    function updateDrainAddress(address newDrainAddress) external;
    function updateOwner(address newOwner) external;
    function updatePool(uint256 poolId, uint256 amount) external;
    function addBulk(ITarget target, address adapter, uint256[] memory targetPids) external;
    function claim(uint256 poolId, address to) external;
    function updateTargetInfo(uint256 poolId, ITarget newTarget, address newAdapter, uint256 newTargetPoolId, bool restake) external;
    function massUpdateTarget(uint256[] memory poolId, ITarget[] memory newTarget, address[] memory newAdapter, uint256[] memory newTargetPoolId, bool[] memory restake) external;
    function massUpdateAdapter(uint256[] memory poolId, address newAdapter) external;
    function deposit(uint256 poolId, uint256 rewardableAmount, bool claimRewards) external;
    function withdraw(uint256 poolId,  uint256 rewardableAmount, bool claimRewards) external;
    function drain(uint256[] memory poolIds) external;
}   

    


// File: Contracts/interfaces/IRewarder.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

interface IRewarder {
    struct UserInfo {
        uint256 rewardableDeposit;
        uint256 lifetimeRewardPerOneEtherOfDeposit;
        uint256 pendingReward;
    }

    struct PoolInfo {
        uint256 lifetimeRewardPerOneEtherOfDeposit;
        uint256 lastRewardBlock;
        uint256 rewardPerBlockPerOneEtherOfDeposit;
        uint256 limitPerBlockPerOneEtherOfDeposit;
    }

    event UpdateUser(address indexed user, uint256 indexed poolId, uint256 shares);
    event Claim(address indexed user, uint256 indexed poolId, uint256 rewardAmount, address to);
    event AddPool(uint256 indexed poolId, uint256 rewards);
    event SetFixedRewardPerBlock(uint256 rewardPerBlock);
    event SetFloatingRewardPerBlock(uint256 rewardRate);
    event UpdatePoolRewards(uint256 indexed poolId, uint256 newRewards);
    event UpdatePoolsRewards(uint256[] poolIds, uint256[] newRewards);
    event UpdatePoolLimit(uint256 indexed poolId, uint256 newLimit);
    event UpdatePool(uint256 indexed poolId, uint256 lastRewardBlock, uint256 totalShares, uint256 lifetimeRewardPerExashare); 
    event UpdateDev(address newDev); 
    event UpdateDao(address newDao);
    function poolCount() external view returns (uint256 pools);
    function pendingReward(uint256 poolId, address userAddress) external view returns (uint256 _pendingReward);

    function updateUser(uint256 poolId, address user, uint256 rewardableDeposit) external;
    function claim(uint256 poolId, address user, address to) external;
    function addBulk(uint256[] memory rewardsPerBlockPerOneEther, uint256[] memory poolIdList, uint256[] memory poolLimits) external;
    function updatePoolRewards(uint256 poolId, uint256 newRewardPerBlockPerEther) external;
    function massUpdatePoolRewards(uint256[] memory poolIds, uint256[] memory newRewardsPerBlockPerEther) external;
    function updatePoolLimit(uint256 poolId, uint256 newLimitPerBlockPerEther) external;
    function massUpdatePoolLimits(uint256[] memory poolIds, uint256[] memory newLimitsPerBlockPerEther) external;
    function massUpdatePools(uint256[] calldata poolIdList) external;
    function updatePool(uint256 poolId) external returns (PoolInfo memory pool, uint256 rewardableDeposits);

    function distributeFee() external;
    function updateDev(address newDev) external;
    function updateDao(address newDao) external;
    
}

// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/token/ERC20/utils/SafeERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (token/ERC20/utils/SafeERC20.sol)

pragma solidity ^0.8.0;

import "../IERC20.sol";
import "../extensions/draft-IERC20Permit.sol";
import "../../../utils/Address.sol";

/**
 * @title SafeERC20
 * @dev Wrappers around ERC20 operations that throw on failure (when the token
 * contract returns false). Tokens that return no value (and instead revert or
 * throw on failure) are also supported, non-reverting calls are assumed to be
 * successful.
 * To use this library you can add a `using SafeERC20 for IERC20;` statement to your contract,
 * which allows you to call the safe operations as `token.safeTransfer(...)`, etc.
 */
library SafeERC20 {
    using Address for address;

    function safeTransfer(
        IERC20 token,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transfer.selector, to, value));
    }

    function safeTransferFrom(
        IERC20 token,
        address from,
        address to,
        uint256 value
    ) internal {
        _callOptionalReturn(token, abi.encodeWithSelector(token.transferFrom.selector, from, to, value));
    }

    /**
     * @dev Deprecated. This function has issues similar to the ones found in
     * {IERC20-approve}, and its usage is discouraged.
     *
     * Whenever possible, use {safeIncreaseAllowance} and
     * {safeDecreaseAllowance} instead.
     */
    function safeApprove(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        // safeApprove should only be called when setting an initial allowance,
        // or when resetting it to zero. To increase and decrease it, use
        // 'safeIncreaseAllowance' and 'safeDecreaseAllowance'
        require(
            (value == 0) || (token.allowance(address(this), spender) == 0),
            "SafeERC20: approve from non-zero to non-zero allowance"
        );
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, value));
    }

    function safeIncreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        uint256 newAllowance = token.allowance(address(this), spender) + value;
        _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
    }

    function safeDecreaseAllowance(
        IERC20 token,
        address spender,
        uint256 value
    ) internal {
        unchecked {
            uint256 oldAllowance = token.allowance(address(this), spender);
            require(oldAllowance >= value, "SafeERC20: decreased allowance below zero");
            uint256 newAllowance = oldAllowance - value;
            _callOptionalReturn(token, abi.encodeWithSelector(token.approve.selector, spender, newAllowance));
        }
    }

    function safePermit(
        IERC20Permit token,
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) internal {
        uint256 nonceBefore = token.nonces(owner);
        token.permit(owner, spender, value, deadline, v, r, s);
        uint256 nonceAfter = token.nonces(owner);
        require(nonceAfter == nonceBefore + 1, "SafeERC20: permit did not succeed");
    }

    /**
     * @dev Imitates a Solidity high-level call (i.e. a regular function call to a contract), relaxing the requirement
     * on the return value: the return value is optional (but if data is returned, it must not be false).
     * @param token The token targeted by the call.
     * @param data The call data (encoded using abi.encode or one of its variants).
     */
    function _callOptionalReturn(IERC20 token, bytes memory data) private {
        // We need to perform a low level call here, to bypass Solidity's return data size checking mechanism, since
        // we're implementing it ourselves. We use {Address.functionCall} to perform this call, which verifies that
        // the target address contains contract code and also asserts for success in the low-level call.

        bytes memory returndata = address(token).functionCall(data, "SafeERC20: low-level call failed");
        if (returndata.length > 0) {
            // Return data is optional
            require(abi.decode(returndata, (bool)), "SafeERC20: ERC20 operation did not succeed");
        }
    }
}

// File: Contracts/libraries/Adapter.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

import { IERC20 } from "@openzeppelin/contracts/interfaces/IERC20.sol";
import { ITarget } from "../interfaces/ITarget.sol";

library Adapter {
    // Pool info
    function lockableToken(ITarget target, uint256 poolId) external view returns (IERC20) {
        (bool success, bytes memory result) = address(target).staticcall(abi.encodeWithSignature("lockableToken(uint256)", poolId));
        require(success, "lockableToken(uint256 poolId) staticcall failed.");
        return abi.decode(result, (IERC20));
    }

    function lockedAmount(ITarget target, uint256 poolId) external view returns (uint256) {
        // note the impersonation
        (bool success, bytes memory result) = address(target).staticcall(abi.encodeWithSignature("lockedAmount(address,uint256)", address(this), poolId));
        require(success, "lockedAmount(uint256 poolId) staticcall failed.");
        return abi.decode(result, (uint256));
    }

    // Pool actions
    function deposit(ITarget target, uint256 poolId, uint256 amount) external {
        (bool success,) = address(target).delegatecall(abi.encodeWithSignature("deposit(address,uint256,uint256)", address(target), poolId, amount));
        require(success, "deposit(uint256 poolId, uint256 amount) delegatecall failed.");
    }

    function withdraw(ITarget target, uint256 poolId, uint256 amount) external {
        (bool success,) = address(target).delegatecall(abi.encodeWithSignature("withdraw(address,uint256,uint256)", address(target), poolId, amount));
        require(success, "withdraw(uint256 poolId, uint256 amount) delegatecall failed.");
    }

    function claimReward(ITarget target, uint256 poolId) external {
        // note the impersonation
        (bool success,) = address(target).delegatecall(abi.encodeWithSignature("claimReward(address,address,uint256)", address(target), address(this), poolId));
        require(success, "claimReward(address, address user, uint256 poolId) delegatecall failed.");
    }

    function poolUpdate(ITarget target, uint256 poolId, uint256 amount) external {
        (bool success,) = address(target).delegatecall(abi.encodeWithSignature("poolUpdate(address,uint256,uint256)", address(target), poolId, amount));
        require(success, "poolUpdate(uint256 poolId, uint256 amount) delegatecall failed.");
    }

}
// File: Contracts/libraries/SafeRewarder.sol



// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

interface Rewarder {}

library SafeRewarder {
    event UpdateUserFailed(Rewarder rewarder, uint256 poolId, address user,  uint256 rewardableDeposit, string reason);
    event ClaimFailed(Rewarder rewarder, uint256 poolId, address user, address to, string reason);

    function updateUser(Rewarder rewarder, uint256 poolId, address user, uint256 rewardableDeposit) internal {
        (bool success, bytes memory returndata) = address(rewarder).call(abi.encodeWithSignature("updateUser(uint256,address,uint256)", poolId, user, rewardableDeposit));
        if(!success) {
            emit UpdateUserFailed(rewarder, poolId, user,  rewardableDeposit, string(returndata));
        }
    }

    function claim(Rewarder rewarder, uint256 poolId, address user, address to) internal {
        (bool success, bytes memory returndata) = address(rewarder).call(abi.encodeWithSignature("claim(uint256,address,address)", poolId, user, to));
        if(!success) {
            emit ClaimFailed(rewarder, poolId, user, to, string(returndata));
        }
    }
}
// File: Contracts/interfaces/ITarget.sol



// SPDX-License-Identifier: MIT
pragma solidity ^0.8.0;

interface ITarget {
    function poolAddress(uint256) external view returns(address);
}
// File: @openzeppelin/contracts/interfaces/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (interfaces/IERC20.sol)

pragma solidity ^0.8.0;

import "../token/ERC20/IERC20.sol";

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.6.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(
        address from,
        address to,
        uint256 amount
    ) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/Address.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.7.0) (utils/Address.sol)

pragma solidity ^0.8.1;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     *
     * [IMPORTANT]
     * ====
     * You shouldn't rely on `isContract` to protect against flash loan attacks!
     *
     * Preventing calls from contracts is highly discouraged. It breaks composability, breaks support for smart wallets
     * like Gnosis Safe, and does not provide security since it can be circumvented by calling from a contract
     * constructor.
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize/address.code.length, which returns 0
        // for contracts in construction, since the code is only stored at the end
        // of the constructor execution.

        return account.code.length > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        (bool success, ) = recipient.call{value: amount}("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain `call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value
    ) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(
        address target,
        bytes memory data,
        uint256 value,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        (bool success, bytes memory returndata) = target.call{value: value}(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        (bool success, bytes memory returndata) = target.staticcall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(
        address target,
        bytes memory data,
        string memory errorMessage
    ) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        (bool success, bytes memory returndata) = target.delegatecall(data);
        return verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Tool to verifies that a low level call was successful, and revert if it wasn't, either by bubbling the
     * revert reason using the provided one.
     *
     * _Available since v4.3._
     */
    function verifyCallResult(
        bool success,
        bytes memory returndata,
        string memory errorMessage
    ) internal pure returns (bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly
                /// @solidity memory-safe-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}

// File: @openzeppelin/contracts/token/ERC20/extensions/draft-IERC20Permit.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (token/ERC20/extensions/draft-IERC20Permit.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 Permit extension allowing approvals to be made via signatures, as defined in
 * https://eips.ethereum.org/EIPS/eip-2612[EIP-2612].
 *
 * Adds the {permit} method, which can be used to change an account's ERC20 allowance (see {IERC20-allowance}) by
 * presenting a message signed by the account. By not relying on {IERC20-approve}, the token holder account doesn't
 * need to send a transaction, and thus is not required to hold Ether at all.
 */
interface IERC20Permit {
    /**
     * @dev Sets `value` as the allowance of `spender` over ``owner``'s tokens,
     * given ``owner``'s signed approval.
     *
     * IMPORTANT: The same issues {IERC20-approve} has related to transaction
     * ordering also apply here.
     *
     * Emits an {Approval} event.
     *
     * Requirements:
     *
     * - `spender` cannot be the zero address.
     * - `deadline` must be a timestamp in the future.
     * - `v`, `r` and `s` must be a valid `secp256k1` signature from `owner`
     * over the EIP712-formatted function arguments.
     * - the signature must use ``owner``'s current nonce (see {nonces}).
     *
     * For more information on the signature format, see the
     * https://eips.ethereum.org/EIPS/eip-2612#specification[relevant EIP
     * section].
     */
    function permit(
        address owner,
        address spender,
        uint256 value,
        uint256 deadline,
        uint8 v,
        bytes32 r,
        bytes32 s
    ) external;

    /**
     * @dev Returns the current nonce for `owner`. This value must be
     * included whenever a signature is generated for {permit}.
     *
     * Every successful call to {permit} increases ``owner``'s nonce by one. This
     * prevents a signature from being used multiple times.
     */
    function nonces(address owner) external view returns (uint256);

    /**
     * @dev Returns the domain separator used in the encoding of the signature for {permit}, as defined by {EIP712}.
     */
    // solhint-disable-next-line func-name-mixedcase
    function DOMAIN_SEPARATOR() external view returns (bytes32);
}